Earth Capture Rate Check

Flip Tanedo
14 April 2017

PREM Import

PREM import, from EarthChecker_v2.nb (Jan 2017):

The data about the density of the Earth comes from the PREM 500 model, available at PREM 500 from http://ds.iris.edu/ds/products/emc-prem/. The citations to this information are:

1. Dziewonski, A.M., and D.L. Anderson. 1981. “Preliminary reference Earth model.” Phys. Earth Plan. Int. 25:297-356.

2. Trabant, C., A. R. Hutko, M. Bahavar, R. Karstens, T. Ahern and R. Aster (2012), Data products at the IRIS DMC: stepping-stones for research and other application, Seismological Research Letters, 83(6), 846:854. doi: 10.1785/0220 120032

We make use of a file **PREM500jordan.csv**, which was processed by Jordan Smolinsky to remove duplicate entries. This file is assumed to be in the same directory as this notebook.

Importing data and key features

In[175]:=
```
myPath = NotebookDirectory[];
PREMlocation = myPath <> "PREM500jordan.csv";

RawPREMdata = Import[PREMlocation, "CSV"];
TempRadius = Transpose[RawPREMdata][[1]] m;
TempDensity = Transpose[RawPREMdata][[2]] kg / m^3;

Radius = TempRadius[[2 ;;]];
Density = TempDensity[[2 ;;]];
PREMlength = Length[Radius];

Rmeters = Radius /. m -> 1;
```

Sanity Check

Shell Thicknesses and Shell Mass

Note that the shells are not evenly spaced in the PREM500 data.

```
In[184]:= ΔR = TempRadius[[2 ;; PREMlength + 1]] - TempRadius[[1 ;; PREMlength]];
ΔRmeters = ΔR /. m → 1;
```

$$\text{ShellMass} = \frac{4.}{3}\pi\,\text{Density}\,\text{Radius}^3 - \frac{4.}{3}\pi\,\text{Density}\,(\text{Radius} - \Delta R)^3;$$

Sanity Check: Earth’s total density

Sanity Check: Plot of Mass in Each Shell

Enclosed Mass

```
In[23]:= EnclosedMass = Accumulate[ShellMass];
```

Check of Enclosed Mass

Escape Velocity

The escape velocity in natural units is given by an integral over the enclosed mass M(s).

$$v^2 = \int_r^\infty \frac{2\,G\,M(s)}{s^2\,c^2}\,ds \quad (*\ [G] = \frac{L^3}{M\,T^2};\ c = \text{speed of light } *)$$

So to do this as an array we break apart the integral into two parts depending on whether one is inside or outside the radius of the Earth, R_{max}.

$$v(r)^2 = \int_r^{R_{max}} \frac{2\,G\,M(s)}{s^2\,c^2}\,ds + \int_{R_{max}}^\infty \frac{2\,G\,M(R_{max})}{s^2\,c^2}\,ds$$

$$= \frac{2\,G}{c^2}\left[\left(\sum_{s=r}^{R_{max}} \frac{M(s)}{s^2}\,\Delta R\right) + \frac{M(R_{max})}{R_{max}}\right]$$

We use the **Accumulate** command to take care of the first term. We don’t actually ask about $r > R_{max}$, so we can ignore the second term.

$$\text{In[26]:=}\quad GN = 6.67384\times 10^{-11}\ \frac{m^3}{kg\ s^2}\ /.\ \{kg \to 1000\ g,\ m \to 100\ cm\};$$

$$c = 3.\times 10^{10}\ \frac{cm}{s};$$

$$v2 = 2\,\frac{GN}{c^2}\Big(\Delta R\,\text{Reverse}\big[\text{Accumulate}\big[\text{Reverse}\big[\text{EnclosedMass}/\text{Radius}^2\big]\big]\big] + \text{EnclosedMass}[[\text{Length}[\text{EnclosedMass}]]] / \text{Radius}[[\text{Length}[\text{Radius}]]]\Big)\ /.\ \{kg \to 1000\ g,\ m \to 100\ cm\};$$

Sanity Checks

Load Data: Crust vs. Mantle

The data for the elemental mass fractions are from https://arxiv.org/abs/astro-ph/0401113:

Element	Atomic number	Mass fraction Core	Mantle
Oxygen, O	16	0.0	0.440
Silicon, Si	28	0.06	0.210
Magnesium, Mg	24	0.0	0.228
Iron, Fe	56	0.855	0.0626
Calcium, Ca	40	0.0	0.0253
Phosphor, P	30	0.002	0.00009
Sodium, Na	23	0.0	0.0027
Sulphur, S	32	0.019	0.00025
Nickel, Ni	59	0.052	0.00196
Aluminum, Al	27	0.0	0.0235
Chromium, Cr	52	0.009	0.0026

In[39]:=
```
Elements = {
   (* symbol, Z, A, Core, Mantle *)
   {"O", 8, 16, 0, .44},
   {"Si", 14, 28, .06, .21},
   {"Mg", 12, 24, 0, .228},
   {"Fe", 26, 56, 0.855, .0626},
   {"Ca", 20, 40, 0, .0253},
   {"P", 15, 30, 0.002, .00009},
   {"Na", 11, 23, 0, .0027},
   {"S", 16, 32, 0.019, .00025},
   {"Ni", 28, 59, 0.052, .00196},
   {"Al", 13, 27, 0, .0235},
   {"Cr", 24, 52, 0.009, .0026}
  };
```

Sanity Check

Elemental Distribution

The core--mantle separator is r=3480 km. This corresponds to element 271 of 491.

In[34]:=
```
Rcore = Radius[[271]];
Rearth = Radius[[491]];
```

Reprocess elemental data

Column 5 is a little tricky, but it gives the number density of the element for each radial slice. The expression is:

$$n_i = (\text{Mass Density in Shell})\ (\text{mass fraction of element})\ \frac{1}{\text{mass of element}}\left[\text{convert to } \frac{1}{\text{cm}^3}\right]$$

In[243]:=
```
MyElements = {
    (*1*)#[[1]] (*name*),
    (*2*)#[[2]] (*Z*),
    (*3*)#[[3]] (*A*),
    (*4*)#[[3]] .938272 GeV (* mass in GeV *),
    (*5*) (Density (* density of shell *)
          Join[
           Table[#[[4]], {271}], (* const mass_i/(total mass) for r_i in core *)
           Table[#[[5]], {Length[Radius] - 271}]
           (* const mass_i/(total mass) for r_i in mantle *)
          ]) / (#[[3]] .938272 GeV (*m_proton*)) /. {kg → 1000 g} /.
      {g → 5.62 × 10^23 GeV} /. {m → 100 cm}
    (* list of n_i for each radius: ρ (mass frac)/(mass) = ρ n_i/(total mass density) *)
   } & /@ Elements;

mn = #[[4]] & /@ MyElements;
mnInGeV = mn /. GeV → 1;

numberdensity = #[[5]] & /@ MyElements;
nmeters = numberdensity /. cm → 10^-2;

Z = #[[2]] & /@ MyElements;
```

Additional Discussion (what's going on in column 5?)

Sanity Check

Dark Matter Velocity Distribution

Start with galactic center rest frame velocity distribution. Eq. 17 of 1509.07525 (Earth paper). We'll write all of our velocities in natural units. This simplifies units, but also has the benefit of making the velocity integral go over a finite range, which may help with the numerical integration.

Basic Velocity Distribution

```
In[80]:= vgal = 550 (km/s)/c /. {km → 100 000 cm}; (* galactic escape velocity *)
k = 2.5; (* index parameter *)
u0 = 245 km / s / c /. {km → 100 000 cm};

(* Normalization *)
N0 = NIntegrate[(E^((vgal^2-u^2)/(k u0^2)) - 1)^k 4 π u^2 UnitStep[vgal - u], {u, 0, 1}];
f0[u_] := 1/N0 (E^((vgal^2-u^2)/(k u0^2)) - 1)^k UnitStep[vgal - u]

(* assume full velocity distribution f[u] is the basic one *)
f[u_] := f0[u];
```

Check

In progress: angular averaged velocity distribution

Capture Rate

Discussion

In order to perform the integral in (15), there are a couple of useful things to do:

(1) cancel the w^2dependence in dσ/dEr with the w^2 in the du integral coming from the distortion of the velocity distribution from the Earth's gravitational potential.

(2) pull out common factors of $\epsilon^2\ \alpha_X\ n_X$ from C_{cap} since these are overall prefactors.

The resulting integral is:

$$\left(C^N\right)_{\text{cap}} = \epsilon^2\ \alpha_X\ n_X\ \left(C^N\right)_{\text{cap,red}}$$

$$\left(C^N\right)_{\text{cap,red}} = \int_0^R dr 4\pi r^2\ n_N\ (r) \int_0^c du\ 4\ \pi\ u\ f\ (u) \int_{E_{min}}^{E_{max}} dE_R\ \frac{8\ \pi\ \alpha\ Z[n]^2\ m[n]}{\left(2\ m[n]\ E_R + mA^2\right)^2}\ F[n,\ E_R]^2$$

$$\text{Emin} = \frac{1}{2}\ mX\ \left(w^2 - v_+[r]^2\right) = \frac{1}{2}\ mX\ u^2$$

$$\text{Emax} = \frac{2\ \mu^2}{mn}\ w^2 = 2\ \frac{mn\ mX^2}{(mn + mX)^2}\ \left(u^2 + v_{esc}[r]^2\right)$$

The integrand depends on:
- n=N, the nucleus

- the integration variables
- dark matter mass (in reduced mass)
- dark photon mass

Important remark: not explicitly shown: a Θ function that enforces $E_{max} > E_{min}$.

Notes on units:
The integral over the earth radius is done in meters.
The integral over the DM velocity is done in natural units (dimensionless velocities)
The integral over the recoil energy is done in GeV.

The final expression for $(C^N)_{cap,red}$ is in $1/\text{GeV}^2$. Upon multiplying by n_X this has units of $\frac{1}{\text{GeV}^2\,\text{cm}^3}$ which we convert into $\frac{1}{\text{sec}}$.

Further Discussion: accounting for the range of asymptotic velocities

The integrand contains a factor $(2\,m[n]\,E_R + mA^2)^{-2}$.

Observe that in the integration range, the E_R-dependent term is always subdominant. This is because we are restricted to the range where $E_{max} > E_{min}$. This implies that we are only sampling very small values of the asymptotic dark matter velocity, u.

Thus we may approximate:

$$\frac{1}{(2\,m[n]\,E_R + mA^2)^2} = \frac{1}{mA^4}\left(1 + 2\,\frac{2\,m[n]\,E_R}{mA^2} + O\left(\frac{2\,m[n]\,E_R}{mA^2}\right)^2\right)$$

Similarly, the Helm form factor is negligible because the momentum transfer is always small. This gives us:

$$(C^N)_{cap,red} = \int_0^R dr 4\pi r^2\, n_N(r) \int_0^c du\, 4\,\pi\,u\,f(u) \int_{E_{min}}^{E_{max}} dE_R\, \frac{8\,\pi\,\alpha\,Z[n]^2\,m[n]}{mA^4}$$

$$= \int_0^R dr 4\pi r^2\, n_N(r) \int_0^c du\, 4\,\pi\,u\,f(u) \times \frac{8\,\pi\,\alpha\,Z[n]^2\,m[n]}{mA^4} (E_{max} - E_{min})\,\Theta\,(E_{max} - E_{min})$$

$$(C^N)_{cap} = \int_0^R dr 4\pi r^2\, n_N(r) \int_0^c du\, 4\,\pi\,u\,f(u) \times \frac{8\,\pi\,\alpha\,Z[n]^2\,m[n]}{mA^4} (E_{max} - E_{min})\,\Theta\,(E_{max} - E_{min})$$

Revised capture rate

$$(C^N)_{cap} = \epsilon^2\,\alpha_X\,n_X\,\frac{8\,\pi\,\alpha\,Z[n]^2\,m[n]}{mA^4}$$

$$\int_0^R dr 4\pi r^2\, n_N(r) \int_0^c du\, 4\,\pi\,u\,f(u)\,(E_{max}(u) - E_{min}(u))\,\Theta\,(E_{max} - E_{min})$$

$$\text{Emin} = \frac{1}{2}\,\text{mX}\left(w^2 - v_+[r]^2\right) = \frac{1}{2}\,\text{mX}\,u^2$$

$$\text{Emax} = \frac{2\,\mu^2}{\text{mn}}\,w^2 = 2\,\frac{\text{mn mX}^2}{(\text{mn}+\text{mX})^2}\left(u^2 + v_{\text{esc}}[r]^2\right)$$

We can improve this by setting an upper limit on the velocity integral. Recall that

$$\text{Emin} = a\,u^2$$

$$\text{Emax} = b\,u^2 + c$$

These match at :

$$u_{\max} = \sqrt{\frac{c}{a-b}} = \sqrt{\frac{2\,\frac{\mu^2}{\text{mn}}\,v_{\text{esc}}[r]^2}{\frac{1}{2}\,\text{mX} - 2\,\frac{\mu^2}{\text{mn}}}} = \frac{v_{\text{esc}}[r]}{\sqrt{\frac{1}{4}\,\frac{\text{mX mn}}{\mu^2} - 1}} = \frac{2\,v_{\text{esc}}[r]}{\sqrt{\frac{\text{mX mn}}{\mu^2} - 4}} = \frac{2\,v_{\text{esc}}[r]}{\sqrt{\frac{(\text{mX}+\text{mn})^2}{\text{mX mn}} - 4}}$$

This value of $u_{\max}$ is the upper limit of the velocity integral. Imposing this also obviates the $\Theta(E_{\max} - E_{\min})$ factor.

With this:

$$\left(C^N\right)_{\text{cap}} = \epsilon^2\,\alpha_X\,n_X\,\frac{8\,\pi\,\alpha\,Z[n]^2\,\text{mn}}{\text{mA}^4}\int_0^R dr\,4\pi r^2\,n_N(r)\int_0^{u_{\max}} du\,4\,\pi\,u\,f(u)\,\left(E_{\max}(u) - E_{\min}(u)\right)$$

$$= \epsilon^2\,\alpha_X\,n_X\,\frac{8\,\pi\,\alpha\,Z[n]^2\,\text{mn}}{\text{mA}^4}$$

$$\int_0^R dr\,4\pi r^2\,n_N(r)\int_0^{u_{\max}} du\,4\,\pi\,u\,f(u)\left[2\,\frac{\text{mn mX}^2}{(\text{mn}+\text{mX})^2}\left(u^2 + v_{\text{esc}}[r]^2\right) - \frac{1}{2}\,\text{mX}\,u^2\right]$$

$$= 8\,\pi\,\epsilon^2\,\alpha\,\alpha_X\,n_X\,Z[n]^2\,\frac{\text{mn mX}}{\text{mA}^4}$$

$$\int_0^R dr\,4\pi r^2\,n_N(r)\int_0^{u_{\max}} du\,4\,\pi\,u\,f(u)\left[2\,\frac{\mu^2}{\text{mn mX}}\left(u^2 + v_{\text{esc}}[r]^2\right) - \frac{1}{2}\,u^2\right]$$

In[124]:=

$$\textbf{umax[rindex\_, n\_, mXinGeV\_] := } 2\sqrt{\frac{\text{v2[[rindex]]}}{\frac{(\text{mXinGeV}+\text{mnInGeV[[n]]})^2}{\text{mXinGeV mnInGeV[[n]]}} - 4}}$$

uIntegrand[u_, rindex_, n_, mXinGeV_] :=

$$4\,\pi\,u\,f[u]\left(\left(2\,\frac{\text{mnInGeV[[n]] mXinGeV}}{(\text{mnInGeV[[n]]} + \text{mXinGeV})^2}\right)\left(u^2 + \text{v2[[rindex]]}\right) - \frac{1}{2}\,u^2\right)$$

In[262]:=
```
(* r integral is in meters, ΔR is included; the integrand is dimensionless *)
rIntegrand[rindex_, n_, mXinGeV_] :=
 4 π Rmeters[[rindex]]^2 ΔRmeters[[rindex]] nmeters[[n]][[rindex]]
  NIntegrate[uIntegrand[u, rindex, n, mXinGeV], {u, 0, umax[rindex, n, mXinGeV]}]

rIntegrands[n_, mXinGeV_] := rIntegrand[#, n, mXinGeV] & /@ Range[Length[Radius]]

Ccap[ϵ_, αX_, n_, mXinGeV_, mAinGeV_] :=
 Total[8 π ϵ^2 1/137. αX (0.3/mXinGeV (* 1/cm^3 *)) Z[[n]]^2 (mnInGeV[[n]] mXinGeV)/mAinGeV^4 (* 1/GeV^2 *)
   rIntegrands[n, mXinGeV] (1.98 × 10^-14 (*cm GeV*))^2 (3 × 10^10 (* cm/s *))] 1/sec
```

Testing, takes 10 seconds

Helm Form Factor

Test

In[265]:=
```
Ccap[10^-8, 0.035, 4 (*Fe*), 1000, 1]
```

Out[265]= $\frac{2.48401 \times 10^{8}}{\text{sec}}$